Datasets:
Tasks:
Text Generation
Modalities:
Text
Sub-tasks:
language-modeling
Languages:
English
Size:
100K - 1M
License:
File size: 23,622 Bytes
4365a98 |
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57 58 59 60 61 62 63 64 65 66 67 68 69 70 71 72 73 74 75 76 77 78 79 80 81 82 83 84 85 86 87 88 89 90 91 92 93 94 95 96 97 98 99 100 101 102 103 104 105 106 107 108 109 110 111 112 113 114 115 116 117 118 119 120 121 122 123 124 125 126 127 128 129 130 131 132 133 134 135 136 137 138 139 140 141 142 143 144 145 146 147 148 149 150 151 152 153 154 155 156 157 158 159 160 161 162 163 164 165 166 167 168 169 170 171 172 173 174 175 176 177 178 179 180 181 182 183 184 185 186 187 188 189 190 191 192 193 194 195 196 197 198 199 200 201 202 203 204 205 206 207 208 209 210 211 212 213 214 215 216 217 218 219 220 221 222 223 224 225 226 227 228 229 230 231 232 233 234 235 236 237 238 239 240 241 242 243 244 245 246 247 248 249 250 251 252 253 254 255 256 257 258 259 260 261 262 263 264 265 266 267 268 269 270 271 272 273 274 275 276 277 278 279 280 281 282 283 284 285 286 287 288 289 290 291 292 293 294 295 296 297 298 299 300 301 302 303 304 305 306 307 308 309 310 311 312 313 314 315 316 317 318 319 320 321 322 323 324 325 326 327 328 329 330 331 332 333 334 335 336 337 338 339 340 341 342 343 344 345 346 347 348 349 350 351 352 353 354 355 356 357 358 359 360 361 362 363 364 365 366 367 368 369 370 371 372 373 374 375 376 377 378 379 380 381 382 383 384 385 386 387 388 389 390 391 392 393 394 395 396 397 398 399 400 401 402 403 404 405 406 407 408 409 410 411 412 413 414 415 416 417 418 419 420 421 422 423 424 425 426 427 428 429 430 431 432 433 434 435 436 437 438 439 440 441 442 443 444 445 446 447 448 449 450 451 452 453 454 455 456 457 458 459 460 461 462 463 464 465 466 467 468 469 470 471 472 473 474 475 476 477 478 479 480 481 482 483 484 485 486 487 488 489 490 491 492 493 494 495 496 497 498 499 500 501 502 503 504 505 506 507 508 509 510 511 512 513 514 515 516 517 518 519 520 521 522 523 524 525 526 527 528 529 530 531 532 533 534 535 536 537 538 539 540 541 542 543 544 545 546 547 548 549 550 551 552 553 554 555 556 557 558 559 |
/-
Copyright (c) 2019 Johannes HΓΆlzl. All rights reserved.
Released under Apache 2.0 license as described in the file LICENSE.
Authors: Johannes HΓΆlzl, Patrick Massot, Casper Putz, Anne Baanen
-/
import linear_algebra.multilinear.basis
import linear_algebra.matrix.reindex
import ring_theory.algebra_tower
import tactic.field_simp
import linear_algebra.matrix.nonsingular_inverse
import linear_algebra.matrix.basis
/-!
# Determinant of families of vectors
This file defines the determinant of an endomorphism, and of a family of vectors
with respect to some basis. For the determinant of a matrix, see the file
`linear_algebra.matrix.determinant`.
## Main definitions
In the list below, and in all this file, `R` is a commutative ring (semiring
is sometimes enough), `M` and its variations are `R`-modules, `ΞΉ`, `ΞΊ`, `n` and `m` are finite
types used for indexing.
* `basis.det`: the determinant of a family of vectors with respect to a basis,
as a multilinear map
* `linear_map.det`: the determinant of an endomorphism `f : End R M` as a
multiplicative homomorphism (if `M` does not have a finite `R`-basis, the
result is `1` instead)
* `linear_equiv.det`: the determinant of an isomorphism `f : M ββ[R] M` as a
multiplicative homomorphism (if `M` does not have a finite `R`-basis, the
result is `1` instead)
## Tags
basis, det, determinant
-/
noncomputable theory
open_locale big_operators
open_locale matrix
open linear_map
open submodule
universes u v w
open linear_map matrix set function
variables {R : Type*} [comm_ring R]
variables {M : Type*} [add_comm_group M] [module R M]
variables {M' : Type*} [add_comm_group M'] [module R M']
variables {ΞΉ : Type*} [decidable_eq ΞΉ] [fintype ΞΉ]
variables (e : basis ΞΉ R M)
section conjugate
variables {A : Type*} [comm_ring A]
variables {m n : Type*} [fintype m] [fintype n]
/-- If `R^m` and `R^n` are linearly equivalent, then `m` and `n` are also equivalent. -/
def equiv_of_pi_lequiv_pi {R : Type*} [comm_ring R] [nontrivial R]
(e : (m β R) ββ[R] (n β R)) : m β n :=
basis.index_equiv (basis.of_equiv_fun e.symm) (pi.basis_fun _ _)
namespace matrix
/-- If `M` and `M'` are each other's inverse matrices, they are square matrices up to
equivalence of types. -/
def index_equiv_of_inv [nontrivial A] [decidable_eq m] [decidable_eq n]
{M : matrix m n A} {M' : matrix n m A}
(hMM' : M β¬ M' = 1) (hM'M : M' β¬ M = 1) :
m β n :=
equiv_of_pi_lequiv_pi (to_lin'_of_inv hMM' hM'M)
lemma det_comm [decidable_eq n] (M N : matrix n n A) : det (M β¬ N) = det (N β¬ M) :=
by rw [det_mul, det_mul, mul_comm]
/-- If there exists a two-sided inverse `M'` for `M` (indexed differently),
then `det (N β¬ M) = det (M β¬ N)`. -/
lemma det_comm' [decidable_eq m] [decidable_eq n]
{M : matrix n m A} {N : matrix m n A} {M' : matrix m n A}
(hMM' : M β¬ M' = 1) (hM'M : M' β¬ M = 1) :
det (M β¬ N) = det (N β¬ M) :=
begin
nontriviality A,
-- Although `m` and `n` are different a priori, we will show they have the same cardinality.
-- This turns the problem into one for square matrices, which is easy.
let e := index_equiv_of_inv hMM' hM'M,
rw [β det_minor_equiv_self e, β minor_mul_equiv _ _ _ (equiv.refl n) _, det_comm,
minor_mul_equiv, equiv.coe_refl, minor_id_id]
end
/-- If `M'` is a two-sided inverse for `M` (indexed differently), `det (M β¬ N β¬ M') = det N`.
See `matrix.det_conj` and `matrix.det_conj'` for the case when `M' = Mβ»ΒΉ` or vice versa. -/
lemma det_conj_of_mul_eq_one [decidable_eq m] [decidable_eq n]
{M : matrix m n A} {M' : matrix n m A} {N : matrix n n A}
(hMM' : M β¬ M' = 1) (hM'M : M' β¬ M = 1) :
det (M β¬ N β¬ M') = det N :=
by rw [β det_comm' hM'M hMM', β matrix.mul_assoc, hM'M, matrix.one_mul]
end matrix
end conjugate
namespace linear_map
/-! ### Determinant of a linear map -/
variables {A : Type*} [comm_ring A] [module A M]
variables {ΞΊ : Type*} [fintype ΞΊ]
/-- The determinant of `linear_map.to_matrix` does not depend on the choice of basis. -/
lemma det_to_matrix_eq_det_to_matrix [decidable_eq ΞΊ]
(b : basis ΞΉ A M) (c : basis ΞΊ A M) (f : M ββ[A] M) :
det (linear_map.to_matrix b b f) = det (linear_map.to_matrix c c f) :=
by rw [β linear_map_to_matrix_mul_basis_to_matrix c b c,
β basis_to_matrix_mul_linear_map_to_matrix b c b,
matrix.det_conj_of_mul_eq_one]; rw [basis.to_matrix_mul_to_matrix, basis.to_matrix_self]
/-- The determinant of an endomorphism given a basis.
See `linear_map.det` for a version that populates the basis non-computably.
Although the `trunc (basis ΞΉ A M)` parameter makes it slightly more convenient to switch bases,
there is no good way to generalize over universe parameters, so we can't fully state in `det_aux`'s
type that it does not depend on the choice of basis. Instead you can use the `det_aux_def'` lemma,
or avoid mentioning a basis at all using `linear_map.det`.
-/
def det_aux : trunc (basis ΞΉ A M) β (M ββ[A] M) β* A :=
trunc.lift
(Ξ» b : basis ΞΉ A M,
(det_monoid_hom).comp (to_matrix_alg_equiv b : (M ββ[A] M) β* matrix ΞΉ ΞΉ A))
(Ξ» b c, monoid_hom.ext $ det_to_matrix_eq_det_to_matrix b c)
/-- Unfold lemma for `det_aux`.
See also `det_aux_def'` which allows you to vary the basis.
-/
lemma det_aux_def (b : basis ΞΉ A M) (f : M ββ[A] M) :
linear_map.det_aux (trunc.mk b) f = matrix.det (linear_map.to_matrix b b f) :=
rfl
-- Discourage the elaborator from unfolding `det_aux` and producing a huge term.
attribute [irreducible] linear_map.det_aux
lemma det_aux_def' {ΞΉ' : Type*} [fintype ΞΉ'] [decidable_eq ΞΉ']
(tb : trunc $ basis ΞΉ A M) (b' : basis ΞΉ' A M) (f : M ββ[A] M) :
linear_map.det_aux tb f = matrix.det (linear_map.to_matrix b' b' f) :=
by { apply trunc.induction_on tb, intro b, rw [det_aux_def, det_to_matrix_eq_det_to_matrix b b'] }
@[simp]
lemma det_aux_id (b : trunc $ basis ΞΉ A M) : linear_map.det_aux b (linear_map.id) = 1 :=
(linear_map.det_aux b).map_one
@[simp]
lemma det_aux_comp (b : trunc $ basis ΞΉ A M) (f g : M ββ[A] M) :
linear_map.det_aux b (f.comp g) = linear_map.det_aux b f * linear_map.det_aux b g :=
(linear_map.det_aux b).map_mul f g
section
open_locale classical
-- Discourage the elaborator from unfolding `det` and producing a huge term by marking it
-- as irreducible.
/-- The determinant of an endomorphism independent of basis.
If there is no finite basis on `M`, the result is `1` instead.
-/
@[irreducible] protected def det : (M ββ[A] M) β* A :=
if H : β (s : finset M), nonempty (basis s A M)
then linear_map.det_aux (trunc.mk H.some_spec.some)
else 1
lemma coe_det [decidable_eq M] : β(linear_map.det : (M ββ[A] M) β* A) =
if H : β (s : finset M), nonempty (basis s A M)
then linear_map.det_aux (trunc.mk H.some_spec.some)
else 1 :=
by { ext, unfold linear_map.det,
split_ifs,
{ congr }, -- use the correct `decidable_eq` instance
refl }
end
-- Auxiliary lemma, the `simp` normal form goes in the other direction
-- (using `linear_map.det_to_matrix`)
lemma det_eq_det_to_matrix_of_finset [decidable_eq M]
{s : finset M} (b : basis s A M) (f : M ββ[A] M) :
f.det = matrix.det (linear_map.to_matrix b b f) :=
have β (s : finset M), nonempty (basis s A M),
from β¨s, β¨bβ©β©,
by rw [linear_map.coe_det, dif_pos, det_aux_def' _ b]; assumption
@[simp] lemma det_to_matrix
(b : basis ΞΉ A M) (f : M ββ[A] M) :
matrix.det (to_matrix b b f) = f.det :=
by { haveI := classical.dec_eq M,
rw [det_eq_det_to_matrix_of_finset b.reindex_finset_range, det_to_matrix_eq_det_to_matrix b] }
@[simp] lemma det_to_matrix' {ΞΉ : Type*} [fintype ΞΉ] [decidable_eq ΞΉ]
(f : (ΞΉ β A) ββ[A] (ΞΉ β A)) :
det f.to_matrix' = f.det :=
by simp [β to_matrix_eq_to_matrix']
@[simp] lemma det_to_lin (b : basis ΞΉ R M) (f : matrix ΞΉ ΞΉ R) :
linear_map.det (matrix.to_lin b b f) = f.det :=
by rw [β linear_map.det_to_matrix b, linear_map.to_matrix_to_lin]
/-- To show `P f.det` it suffices to consider `P (to_matrix _ _ f).det` and `P 1`. -/
@[elab_as_eliminator]
lemma det_cases [decidable_eq M] {P : A β Prop} (f : M ββ[A] M)
(hb : β (s : finset M) (b : basis s A M), P (to_matrix b b f).det) (h1 : P 1) :
P f.det :=
begin
unfold linear_map.det,
split_ifs with h,
{ convert hb _ h.some_spec.some,
apply det_aux_def' },
{ exact h1 }
end
@[simp]
lemma det_comp (f g : M ββ[A] M) : (f.comp g).det = f.det * g.det :=
linear_map.det.map_mul f g
@[simp]
lemma det_id : (linear_map.id : M ββ[A] M).det = 1 :=
linear_map.det.map_one
/-- Multiplying a map by a scalar `c` multiplies its determinant by `c ^ dim M`. -/
@[simp] lemma det_smul {π : Type*} [field π] {M : Type*} [add_comm_group M] [module π M]
(c : π) (f : M ββ[π] M) :
linear_map.det (c β’ f) = c ^ (finite_dimensional.finrank π M) * linear_map.det f :=
begin
by_cases H : β (s : finset M), nonempty (basis s π M),
{ haveI : finite_dimensional π M,
{ rcases H with β¨s, β¨hsβ©β©, exact finite_dimensional.of_finset_basis hs },
simp only [β det_to_matrix (finite_dimensional.fin_basis π M), linear_equiv.map_smul,
fintype.card_fin, det_smul] },
{ classical,
have : finite_dimensional.finrank π M = 0 := finrank_eq_zero_of_not_exists_basis H,
simp [coe_det, H, this] }
end
lemma det_zero' {ΞΉ : Type*} [fintype ΞΉ] [nonempty ΞΉ] (b : basis ΞΉ A M) :
linear_map.det (0 : M ββ[A] M) = 0 :=
by { haveI := classical.dec_eq ΞΉ,
rw [β det_to_matrix b, linear_equiv.map_zero, det_zero],
assumption }
/-- In a finite-dimensional vector space, the zero map has determinant `1` in dimension `0`,
and `0` otherwise. We give a formula that also works in infinite dimension, where we define
the determinant to be `1`. -/
@[simp] lemma det_zero {π : Type*} [field π] {M : Type*} [add_comm_group M] [module π M] :
linear_map.det (0 : M ββ[π] M) = (0 : π) ^ (finite_dimensional.finrank π M) :=
by simp only [β zero_smul π (1 : M ββ[π] M), det_smul, mul_one, monoid_hom.map_one]
/-- Conjugating a linear map by a linear equiv does not change its determinant. -/
@[simp] lemma det_conj {N : Type*} [add_comm_group N] [module A N]
(f : M ββ[A] M) (e : M ββ[A] N) :
linear_map.det ((e : M ββ[A] N) ββ (f ββ (e.symm : N ββ[A] M))) = linear_map.det f :=
begin
classical,
by_cases H : β (s : finset M), nonempty (basis s A M),
{ rcases H with β¨s, β¨bβ©β©,
rw [β det_to_matrix b f, β det_to_matrix (b.map e), to_matrix_comp (b.map e) b (b.map e),
to_matrix_comp (b.map e) b b, β matrix.mul_assoc, matrix.det_conj_of_mul_eq_one],
{ rw [β to_matrix_comp, linear_equiv.comp_coe, e.symm_trans_self,
linear_equiv.refl_to_linear_map, to_matrix_id] },
{ rw [β to_matrix_comp, linear_equiv.comp_coe, e.self_trans_symm,
linear_equiv.refl_to_linear_map, to_matrix_id] } },
{ have H' : Β¬ (β (t : finset N), nonempty (basis t A N)),
{ contrapose! H,
rcases H with β¨s, β¨bβ©β©,
exact β¨_, β¨(b.map e.symm).reindex_finset_rangeβ©β© },
simp only [coe_det, H, H', pi.one_apply, dif_neg, not_false_iff] }
end
/-- If a linear map is invertible, so is its determinant. -/
lemma is_unit_det {A : Type*} [comm_ring A] [module A M]
(f : M ββ[A] M) (hf : is_unit f) : is_unit f.det :=
begin
obtain β¨g, hgβ© : β g, f.comp g = 1 := hf.exists_right_inv,
have : linear_map.det f * linear_map.det g = 1,
by simp only [β linear_map.det_comp, hg, monoid_hom.map_one],
exact is_unit_of_mul_eq_one _ _ this,
end
/-- If a linear map has determinant different from `1`, then the space is finite-dimensional. -/
lemma finite_dimensional_of_det_ne_one {π : Type*} [field π] [module π M]
(f : M ββ[π] M) (hf : f.det β 1) : finite_dimensional π M :=
begin
by_cases H : β (s : finset M), nonempty (basis s π M),
{ rcases H with β¨s, β¨hsβ©β©, exact finite_dimensional.of_finset_basis hs },
{ classical,
simp [linear_map.coe_det, H] at hf,
exact hf.elim }
end
/-- If the determinant of a map vanishes, then the map is not onto. -/
lemma range_lt_top_of_det_eq_zero {π : Type*} [field π] [module π M]
{f : M ββ[π] M} (hf : f.det = 0) : f.range < β€ :=
begin
haveI : finite_dimensional π M, by simp [f.finite_dimensional_of_det_ne_one, hf],
contrapose hf,
simp only [lt_top_iff_ne_top, not_not, β is_unit_iff_range_eq_top] at hf,
exact is_unit_iff_ne_zero.1 (f.is_unit_det hf)
end
/-- If the determinant of a map vanishes, then the map is not injective. -/
lemma bot_lt_ker_of_det_eq_zero {π : Type*} [field π] [module π M]
{f : M ββ[π] M} (hf : f.det = 0) : β₯ < f.ker :=
begin
haveI : finite_dimensional π M, by simp [f.finite_dimensional_of_det_ne_one, hf],
contrapose hf,
simp only [bot_lt_iff_ne_bot, not_not, β is_unit_iff_ker_eq_bot] at hf,
exact is_unit_iff_ne_zero.1 (f.is_unit_det hf)
end
end linear_map
namespace linear_equiv
/-- On a `linear_equiv`, the domain of `linear_map.det` can be promoted to `RΛ£`. -/
protected def det : (M ββ[R] M) β* RΛ£ :=
(units.map (linear_map.det : (M ββ[R] M) β* R)).comp
(linear_map.general_linear_group.general_linear_equiv R M).symm.to_monoid_hom
@[simp] lemma coe_det (f : M ββ[R] M) : βf.det = linear_map.det (f : M ββ[R] M) := rfl
@[simp] lemma coe_inv_det (f : M ββ[R] M) : β(f.detβ»ΒΉ) = linear_map.det (f.symm : M ββ[R] M) := rfl
@[simp] lemma det_refl : (linear_equiv.refl R M).det = 1 := units.ext $ linear_map.det_id
@[simp] lemma det_trans (f g : M ββ[R] M) : (f.trans g).det = g.det * f.det := map_mul _ g f
@[simp] lemma det_symm (f : M ββ[R] M) : f.symm.det = f.detβ»ΒΉ := map_inv _ f
/-- Conjugating a linear equiv by a linear equiv does not change its determinant. -/
@[simp] lemma det_conj (f : M ββ[R] M) (e : M ββ[R] M') :
((e.symm.trans f).trans e).det = f.det :=
by rw [βunits.eq_iff, coe_det, coe_det, βcomp_coe, βcomp_coe, linear_map.det_conj]
end linear_equiv
/-- The determinants of a `linear_equiv` and its inverse multiply to 1. -/
@[simp] lemma linear_equiv.det_mul_det_symm {A : Type*} [comm_ring A] [module A M]
(f : M ββ[A] M) : (f : M ββ[A] M).det * (f.symm : M ββ[A] M).det = 1 :=
by simp [βlinear_map.det_comp]
/-- The determinants of a `linear_equiv` and its inverse multiply to 1. -/
@[simp] lemma linear_equiv.det_symm_mul_det {A : Type*} [comm_ring A] [module A M]
(f : M ββ[A] M) : (f.symm : M ββ[A] M).det * (f : M ββ[A] M).det = 1 :=
by simp [βlinear_map.det_comp]
-- Cannot be stated using `linear_map.det` because `f` is not an endomorphism.
lemma linear_equiv.is_unit_det (f : M ββ[R] M') (v : basis ΞΉ R M) (v' : basis ΞΉ R M') :
is_unit (linear_map.to_matrix v v' f).det :=
begin
apply is_unit_det_of_left_inverse,
simpa using (linear_map.to_matrix_comp v v' v f.symm f).symm
end
/-- Specialization of `linear_equiv.is_unit_det` -/
lemma linear_equiv.is_unit_det' {A : Type*} [comm_ring A] [module A M]
(f : M ββ[A] M) : is_unit (linear_map.det (f : M ββ[A] M)) :=
is_unit_of_mul_eq_one _ _ f.det_mul_det_symm
/-- The determinant of `f.symm` is the inverse of that of `f` when `f` is a linear equiv. -/
lemma linear_equiv.det_coe_symm {π : Type*} [field π] [module π M]
(f : M ββ[π] M) : (f.symm : M ββ[π] M).det = (f : M ββ[π] M).det β»ΒΉ :=
by field_simp [is_unit.ne_zero f.is_unit_det']
/-- Builds a linear equivalence from a linear map whose determinant in some bases is a unit. -/
@[simps]
def linear_equiv.of_is_unit_det {f : M ββ[R] M'} {v : basis ΞΉ R M} {v' : basis ΞΉ R M'}
(h : is_unit (linear_map.to_matrix v v' f).det) : M ββ[R] M' :=
{ to_fun := f,
map_add' := f.map_add,
map_smul' := f.map_smul,
inv_fun := to_lin v' v (to_matrix v v' f)β»ΒΉ,
left_inv := Ξ» x,
calc to_lin v' v (to_matrix v v' f)β»ΒΉ (f x)
= to_lin v v ((to_matrix v v' f)β»ΒΉ β¬ to_matrix v v' f) x :
by { rw [to_lin_mul v v' v, to_lin_to_matrix, linear_map.comp_apply] }
... = x : by simp [h],
right_inv := Ξ» x,
calc f (to_lin v' v (to_matrix v v' f)β»ΒΉ x)
= to_lin v' v' (to_matrix v v' f β¬ (to_matrix v v' f)β»ΒΉ) x :
by { rw [to_lin_mul v' v v', linear_map.comp_apply, to_lin_to_matrix v v'] }
... = x : by simp [h] }
@[simp] lemma linear_equiv.coe_of_is_unit_det {f : M ββ[R] M'} {v : basis ΞΉ R M} {v' : basis ΞΉ R M'}
(h : is_unit (linear_map.to_matrix v v' f).det) :
(linear_equiv.of_is_unit_det h : M ββ[R] M') = f :=
by { ext x, refl }
/-- Builds a linear equivalence from a linear map on a finite-dimensional vector space whose
determinant is nonzero. -/
@[reducible] def linear_map.equiv_of_det_ne_zero
{π : Type*} [field π] {M : Type*} [add_comm_group M] [module π M]
[finite_dimensional π M] (f : M ββ[π] M) (hf : linear_map.det f β 0) :
M ββ[π] M :=
have is_unit (linear_map.to_matrix (finite_dimensional.fin_basis π M)
(finite_dimensional.fin_basis π M) f).det :=
by simp only [linear_map.det_to_matrix, is_unit_iff_ne_zero.2 hf],
linear_equiv.of_is_unit_det this
lemma linear_map.associated_det_of_eq_comp (e : M ββ[R] M) (f f' : M ββ[R] M)
(h : β x, f x = f' (e x)) : associated f.det f'.det :=
begin
suffices : associated (f' ββ βe).det f'.det,
{ convert this using 2, ext x, exact h x },
rw [β mul_one f'.det, linear_map.det_comp],
exact associated.mul_left _ (associated_one_iff_is_unit.mpr e.is_unit_det')
end
lemma linear_map.associated_det_comp_equiv {N : Type*} [add_comm_group N] [module R N]
(f : N ββ[R] M) (e e' : M ββ[R] N) :
associated (f ββ βe).det (f ββ βe').det :=
begin
refine linear_map.associated_det_of_eq_comp (e.trans e'.symm) _ _ _,
intro x,
simp only [linear_map.comp_apply, linear_equiv.coe_coe, linear_equiv.trans_apply,
linear_equiv.apply_symm_apply],
end
/-- The determinant of a family of vectors with respect to some basis, as an alternating
multilinear map. -/
def basis.det : alternating_map R M R ΞΉ :=
{ to_fun := Ξ» v, det (e.to_matrix v),
map_add' := begin
intros v i x y,
simp only [e.to_matrix_update, linear_equiv.map_add],
apply det_update_column_add
end,
map_smul' := begin
intros u i c x,
simp only [e.to_matrix_update, algebra.id.smul_eq_mul, linear_equiv.map_smul],
apply det_update_column_smul
end,
map_eq_zero_of_eq' := begin
intros v i j h hij,
rw [βfunction.update_eq_self i v, h, βdet_transpose, e.to_matrix_update,
βupdate_row_transpose, βe.to_matrix_transpose_apply],
apply det_zero_of_row_eq hij,
rw [update_row_ne hij.symm, update_row_self],
end }
lemma basis.det_apply (v : ΞΉ β M) : e.det v = det (e.to_matrix v) := rfl
lemma basis.det_self : e.det e = 1 :=
by simp [e.det_apply]
/-- `basis.det` is not the zero map. -/
lemma basis.det_ne_zero [nontrivial R] : e.det β 0 :=
Ξ» h, by simpa [h] using e.det_self
lemma is_basis_iff_det {v : ΞΉ β M} :
linear_independent R v β§ span R (set.range v) = β€ β is_unit (e.det v) :=
begin
split,
{ rintro β¨hli, hspanβ©,
set v' := basis.mk hli hspan.ge with v'_eq,
rw e.det_apply,
convert linear_equiv.is_unit_det (linear_equiv.refl _ _) v' e using 2,
ext i j,
simp },
{ intro h,
rw [basis.det_apply, basis.to_matrix_eq_to_matrix_constr] at h,
set v' := basis.map e (linear_equiv.of_is_unit_det h) with v'_def,
have : β v' = v,
{ ext i, rw [v'_def, basis.map_apply, linear_equiv.of_is_unit_det_apply, e.constr_basis] },
rw β this,
exact β¨v'.linear_independent, v'.span_eqβ© },
end
lemma basis.is_unit_det (e' : basis ΞΉ R M) : is_unit (e.det e') :=
(is_basis_iff_det e).mp β¨e'.linear_independent, e'.span_eqβ©
/-- Any alternating map to `R` where `ΞΉ` has the cardinality of a basis equals the determinant
map with respect to that basis, multiplied by the value of that alternating map on that basis. -/
lemma alternating_map.eq_smul_basis_det (f : alternating_map R M R ΞΉ) : f = f e β’ e.det :=
begin
refine basis.ext_alternating e (Ξ» i h, _),
let Ο : equiv.perm ΞΉ := equiv.of_bijective i (fintype.injective_iff_bijective.1 h),
change f (e β Ο) = (f e β’ e.det) (e β Ο),
simp [alternating_map.map_perm, basis.det_self]
end
@[simp] lemma alternating_map.map_basis_eq_zero_iff (f : alternating_map R M R ΞΉ) :
f e = 0 β f = 0 :=
β¨Ξ» h, by simpa [h] using f.eq_smul_basis_det e, Ξ» h, h.symm βΈ alternating_map.zero_apply _β©
lemma alternating_map.map_basis_ne_zero_iff (f : alternating_map R M R ΞΉ) :
f e β 0 β f β 0 :=
not_congr $ f.map_basis_eq_zero_iff e
variables {A : Type*} [comm_ring A] [module A M]
@[simp] lemma basis.det_comp (e : basis ΞΉ A M) (f : M ββ[A] M) (v : ΞΉ β M) :
e.det (f β v) = f.det * e.det v :=
by { rw [basis.det_apply, basis.det_apply, β f.det_to_matrix e, β matrix.det_mul,
e.to_matrix_eq_to_matrix_constr (f β v), e.to_matrix_eq_to_matrix_constr v,
β to_matrix_comp, e.constr_comp] }
lemma basis.det_reindex {ΞΉ' : Type*} [fintype ΞΉ'] [decidable_eq ΞΉ']
(b : basis ΞΉ R M) (v : ΞΉ' β M) (e : ΞΉ β ΞΉ') :
(b.reindex e).det v = b.det (v β e) :=
by rw [basis.det_apply, basis.to_matrix_reindex', det_reindex_alg_equiv, basis.det_apply]
lemma basis.det_reindex_symm {ΞΉ' : Type*} [fintype ΞΉ'] [decidable_eq ΞΉ']
(b : basis ΞΉ R M) (v : ΞΉ β M) (e : ΞΉ' β ΞΉ) :
(b.reindex e.symm).det (v β e) = b.det v :=
by rw [basis.det_reindex, function.comp.assoc, e.self_comp_symm, function.comp.right_id]
@[simp]
lemma basis.det_map (b : basis ΞΉ R M) (f : M ββ[R] M') (v : ΞΉ β M') :
(b.map f).det v = b.det (f.symm β v) :=
by { rw [basis.det_apply, basis.to_matrix_map, basis.det_apply] }
lemma basis.det_map' (b : basis ΞΉ R M) (f : M ββ[R] M') :
(b.map f).det = b.det.comp_linear_map f.symm :=
alternating_map.ext $ b.det_map f
@[simp] lemma pi.basis_fun_det : (pi.basis_fun R ΞΉ).det = matrix.det_row_alternating :=
begin
ext M,
rw [basis.det_apply, basis.coe_pi_basis_fun.to_matrix_eq_transpose, det_transpose],
end
/-- If we fix a background basis `e`, then for any other basis `v`, we can characterise the
coordinates provided by `v` in terms of determinants relative to `e`. -/
lemma basis.det_smul_mk_coord_eq_det_update {v : ΞΉ β M}
(hli : linear_independent R v) (hsp : β€ β€ span R (range v)) (i : ΞΉ) :
(e.det v) β’ (basis.mk hli hsp).coord i = e.det.to_multilinear_map.to_linear_map v i :=
begin
apply (basis.mk hli hsp).ext,
intros k,
rcases eq_or_ne k i with rfl | hik;
simp only [algebra.id.smul_eq_mul, basis.coe_mk, linear_map.smul_apply, linear_map.coe_mk,
multilinear_map.to_linear_map_apply],
{ rw [basis.mk_coord_apply_eq, mul_one, update_eq_self], congr, },
{ rw [basis.mk_coord_apply_ne hik, mul_zero, eq_comm],
exact e.det.map_eq_zero_of_eq _ (by simp [hik, function.update_apply]) hik, },
end
/-- The determinant of a basis constructed by `units_smul` is the product of the given units. -/
@[simp] lemma basis.det_units_smul (w : ΞΉ β RΛ£) : e.det (e.units_smul w) = β i, w i :=
by simp [basis.det_apply]
/-- The determinant of a basis constructed by `is_unit_smul` is the product of the given units. -/
@[simp] lemma basis.det_is_unit_smul {w : ΞΉ β R} (hw : β i, is_unit (w i)) :
e.det (e.is_unit_smul hw) = β i, w i :=
e.det_units_smul _
|